

Mathematische Methoden in der Physik

Bonus Unterlagen : Aufgaben und Lösungen

Hinweis : Diese Texte wurden mit Hilfe von WOLFRAM MATHEMATICA 12 (Wolfram Research, Inc.) erstellt.

Viele in diesen Aufgaben gestellten Probleme können mit *Mathematica* direkt durch spezielle Befehle (wie etwa `Limit` oder `Series`) gelöst werden. Da es uns hier aber auf das entsprechende Hintergrundwissen ankommt, werden wir meist ausführlicher vorgehen und erst in späteren Rechnungen oder zur Kontrolle diese Befehle verwenden.

Kapitel 2. Komplexe Zahlen

2.1

Zeichnen Sie die folgenden Zahlen in der komplexen Ebene

$$(1 + i), (1 - i), (1 + i)^2, (i + \sqrt{3})^2, \\ 2 \left(\cos \frac{\pi}{4} + i \sin \frac{\pi}{4} \right), 4 \left(\cos \frac{4\pi}{3} + i \sin \frac{4\pi}{3} \right), \\ 2 \exp(-3i\pi/2),$$

und geben Sie für jede drei verschiedene Notationen an.

Lösungsweg

Es geht um folgende komplexe Ausdrücke:

```
In[ ]:= zPunkte = {1 + I, 1 - I, (1 + I)^2, (I + Sqrt[3])^2, \\ 2 (Cos [π/4] + I Sin [π/4]), 4 (Cos [4π/3] + I Sin [4π/3]), 2 E^(-3 I π/2)}
```

```
Out[ ]:= {1 + i, 1 - i, 2 i, (i + √3)^2, (1 + i) √2, 4 (-1/2 - i√3/2), 2 i}
```

1. Darstellung: Wir berechnen zuerst Realteil und Imaginärteil:

```
In[ ]:= zPunkte // ComplexExpand
```

```
Out[ ]:= {1 + i, 1 - i, 2 i, 2 + 2 i √3, (1 + i) √2, -2 - 2 i √3, 2 i}
```

2. Darstellung durch Betrag und Argument:

In[*]:= **rWerte = Abs[zPunkte]**

Out[*]:= $\{\sqrt{2}, \sqrt{2}, 2, 4, 2, 4, 2\}$

In[*]:= **phiWerte = Arg[zPunkte]**

Out[*]:= $\{\frac{\pi}{4}, -\frac{\pi}{4}, \frac{\pi}{2}, \frac{\pi}{3}, \frac{\pi}{4}, -\frac{2\pi}{3}, \frac{\pi}{2}\}$


..jeweils als

(r, φ) oder $r \exp(i\varphi)$.

Graphische Darstellung:

In[*]:= **PlotPunkte = Transpose[{Re[zPunkte], Im[zPunkte]}];**

In[*]:= **ListPlot[PlotPunkte, PlotStyle -> PointSize[0.03]]**



Suchbild: Welcher Punkt entspricht welchem z-Wert?

In[*]:= **ClearAll["Global`*"];**

2.2

Lösen Sie die Gleichungen:

(a) $z = (2 + 3i) \bar{z}$

(b) $z = -i \bar{z}$

(c) $z^2 = \bar{z}^2$

(d) $\frac{z + 2 + 3i}{2z - 3} = i + 2$

(e) $\frac{z}{1+i} - \frac{z}{1-i} = 1 + (z - \bar{z}) \sin(\pi + i \ln 3)$

(f) $-2iz = \frac{1 + \bar{z}}{1+i}$

Lösungsweg

Alle diese Gleichungen entsprechen eigentlich jeweils zwei Gleichungen: eine für den reellen Teil und eine für den imaginären. Am einfachsten ist es daher meist, für $z = x + iy$ zu setzen (dabei werden natürlich x und y reell angenommen) und die Gleichung in einen reellen und imaginären Teil aufzuspalten, die beide verschwinden müssen. Das ergibt die beiden Gleichungen. Die

Lösung ist meist keine Punktlösung sondern eine Punktmenge.

$$(a) z = (2 + 3i) \bar{z}$$

Wir teilen die rechte Seite in Real- und Imaginärteil auf:

```
In[ ]:= Re[(2 + 3 I) (x - I y)] // ComplexExpand
```

```
Out[ ]:= 2 x + 3 y
```

```
In[ ]:= Im[(2 + 3 I) (x - I y)] // ComplexExpand
```

```
Out[ ]:= 3 x - 2 y
```

Nun lösen wir das sich ergebende Gleichungssystem:

```
In[ ]:= Solve[{x == 2 x + 3 y, y == 3 x - 2 y}, {x, y}]
```

```
Out[ ]:= {{x -> 0, y -> 0}}
```

Die Lösung ist also $z=0$.

$$(b) z = -i \bar{z}$$

Wir sehen:

$$x + iy = -i(x - iy)$$

und daher

$$x + iy = -y - ix \rightarrow$$

$$x = -y, \quad y = -x$$

Die Lösung ist daher eben diese Gerade $\operatorname{Re} z = -\operatorname{Im} z$ in der komplexen Ebene.

Auch in polarer Form ist diese Gleichung zu lösen:

$$r e^{i\varphi} = -i r e^{-i\varphi}$$

$$\text{oder, da } -i = e^{i3\pi/2+2in\pi}:$$

$$r e^{i\varphi} = r e^{i(3/2+2n)\pi-i\varphi}$$

Die Lösung ist also für beliebige r :

$$\varphi = (3/2 + 2n)\pi - \varphi \rightarrow \varphi = \frac{3\pi}{4} + n\pi$$

Auch diese Form beschreibt dieselbe Gerade.

$$(c) z^2 = \bar{z}^2$$

In Polarkoordinaten:

$$r^2 e^{2i\varphi} = r^2 e^{-2i\varphi+2n\pi i}$$

$$\rightarrow 2\varphi = 2n\pi - 2\varphi$$

$$\rightarrow \varphi = n\pi / 2 \quad (n = 0, 1)$$

unabhängig von r . Die Lösungsmenge enthält also die x -Achse und die y -Achse.

```
In[ ]:= ClearAll["Global`*"];
```

2.3

Die Bahn eines Teilchens liegt in einer Ebene und kann in Abhängigkeit von der Zeit t durch $z = 1 + 2 \exp(4it)$ parametrisiert werden. Berechnen Sie Geschwindigkeit und Beschleunigung.

Welche Form hat die Bahn? Wohin zeigen Geschwindigkeit und Beschleunigung bei $t=0$ und bei $t=\pi/8$?

Lösungsweg

Es handelt sich um eine periodische Bahn, da die Position bei $t=0$ ($z=3$ entsprechend $(x,y)=(3,0)$) zur Zeit $t=\pi/2$ wieder durchlaufen wird.

```
In[ ]:= z = 1 + 2 Exp[4 I t];
```


```
In[ ]:= Re[z] // ComplexExpand
```

```
Out[ ]:= 1 + 2 Cos[4 t]
```

```
In[ ]:= Im[z] // ComplexExpand
```

```
Out[ ]:= 2 Sin[4 t]
```

```
In[ ]:= ParametricPlot[{1 + 2 Cos[4 t], 2 Sin[4 t]}, {t, 0, Pi/2}]
```



Die Geschwindigkeit ist die Ableitung nach t :

```
In[ ]:= v = D[z, t]
```

```
Out[ ]:= 8 i e^{4 i t}
```

```
In[ ]:= v // ComplexExpand
```

```
Out[ ]:= 8 i Cos[4 t] - 8 Sin[4 t]
```

```
In[ ]:= v /. t -> 0
```

```
Out[ ]:= 8 i
```

Die Komponenten von v bei $t=0$ sind daher $(0,8)$.

```
In[ ]:= a = D[v, t]
```

```
Out[ ]:= -32 e4 i t
```

```
In[ ]:= a // ComplexExpand
```

```
Out[ ]:= -32 Cos[4 t] - 32 i Sin[4 t]
```

```
In[ ]:= a /. t -> 0
```

```
Out[ ]:= -32
```

Die Komponenten von a bei $t=0$ sind daher $(-32,0)$. Wie üblich bei solchen Bahnkurven steht die Beschleunigung senkrecht auf die Geschwindigkeit.

Bei $t=\pi/8$ ergibt sich:

```
In[ ]:= v /. t -> Pi / 8
```

```
Out[ ]:= -8
```

```
In[ ]:= a /. t -> Pi / 8
```

```
Out[ ]:= -32 i
```

```
In[ ]:= ClearAll["Global`*"];
```

2.4

Berechnen Sie $\sin(\pi/2 + i \ln 2)$.

Lösungsweg

Wir drücken die Winkelfunktion durch Exponentialfunktionen aus:

$$\begin{aligned} & \frac{1}{2i} (\text{Exp}[i(\pi/2 + i \ln 2)] \\ & \quad - \text{Exp}[-i(\pi/2 + i \ln 2)]) = \\ & \frac{1}{2i} (\text{Exp}[i\pi/2 - \ln 2] \\ & \quad - \text{Exp}[-i\pi/2 + \ln 2]) = \\ & \frac{1}{2i} (\text{Exp}[i\pi/2] / 2 - 2 \text{Exp}[-i\pi/2]) = \\ & \frac{1}{2i} (i/2 + 2i) = \frac{1}{4} + 1 = \frac{5}{4} \end{aligned}$$

```
In[ ]:= ClearAll["Global`*"];
```

2.5

Berechnen Sie mit Hilfe von (2.17) den Betrag, sowie die Real- und Imaginärteile von

(a) $\frac{2i - 1}{i - 2}$

(b) $(1 + 2i)^3$

(c) $\frac{3i}{i - \sqrt{3}}$

Lösungsweg

(2.17) besagt: $\operatorname{Re}(z) = (z + \bar{z})/2$, $\operatorname{Im}(z) = (z - \bar{z})/2i$, $|z| = \sqrt{z\bar{z}}$

(a)

$$\begin{aligned} 2 \operatorname{Re}(z) &= \frac{2i - 1}{i - 2} + \frac{-2i - 1}{-i - 2} \\ &= \frac{(2i - 1)(-i - 2) + (-2i - 1)(i - 2)}{(i - 2)(-i - 2)} \\ &= \frac{(2i - 1)(-i - 2) + (-2i - 1)(i - 2)}{(i - 2)(-i - 2)} = \frac{8}{5} \end{aligned}$$

und daher $\operatorname{Re}(z) = 8/10$. Analog ergibt sich $\operatorname{Im}(z) = -6/10$.

Der Betrag ist

$$\begin{aligned} (|z|)^2 = z\bar{z} &= \frac{2i - 1}{i - 2} \frac{-2i - 1}{-i - 2} \\ &= \frac{1 + 4}{1 + 4} = 1. \end{aligned}$$

(b)

Der Rechengang ist wie bei Aufgabe (a).

Wir berechnen hier nur den Betrag:

$$\begin{aligned} (|z|)^2 = z\bar{z} &= (1 + 2i)^3 (1 - 2i)^3 \\ &= ((1 + 2i)(1 - 2i))^3 \\ &= (1 + 4)^3 = 125 \rightarrow \\ |z| &= 5\sqrt{5} \end{aligned}$$

`In[]:= ClearAll["Global`*"];`

2.8

Welche Figur bilden die Punktmengen, für die gilt:

- (a) $|z| \leq 3$
- (b) $|z + 1| + |z - 1| = 8$
- (c) $\operatorname{Im} z > \operatorname{Re} z$
- (d) $\operatorname{Im} z + \operatorname{Re} z = 1$
- (e) $0 \leq \operatorname{Arg} z < \frac{\pi}{2}$
- (f) $|z - 1 + i| = 4$
- (g) $z = -\bar{z}$
- (h) $1 < (z - 1)(\bar{z} - 1) < 2$
- (i) $0 < z + \bar{z} < 1$

Lösungsweg

Man ersetze einfach z durch $x + iy$ um die Gleichungen der Grenzkurven zu erhalten.

(a) Abgeschlossene Kreisscheibe, Mittelpunkt (0,0), Radius 3;

(b) Ellipse;

(c) $y > x$ ergibt die obere halbe Ebene, über der Grenzgeraden $y = x$.

(d) $y + x = 1$ beschreibt eine Gerade.

(e) (Im Buch ist die untere Grenze nicht angegeben, da implizit angenommen wird, dass es sich um den Hauptwert handelt.) Es handelt sich um ein Viertelkreissegment, ein Rand ist abgeschlossen, einer offen.

(f) Kreisrand;

(g) $z = -\bar{z}$ ergibt $x + iy = -x + iy$

also $x = -x$, daher $x = 0$, y beliebig: die imaginäre Achse.

(h) $(x - 1 + iy)(x - 1 - iy) = (x - 1)^2 + y^2$

Die beiden Grenzkurven sind daher Kreise um den Punkt $(1, 0)$ mit $r = 1$ und $\sqrt{2}$.

(i) Streifen $0 < |x| < 1/2$.

```
In[ ]:= ClearAll["Global`*"];
```

2.9

Zeigen Sie aus der Exponentialdarstellung und aus der Definition durch eine Potenzreihe, dass auch für komplexe Argumente gilt:

(a) $\frac{d}{dz} \sin z = \cos z$

(b) $\cosh^2 z - \sinh^2 z = 1$

Lösungsweg

Exponentialdarstellung:

(a)

$$\begin{aligned} \frac{d}{dz} (e^{iz} - e^{-iz}) \frac{1}{2i} &= (ie^{iz} + ie^{-iz}) \frac{1}{2i} \\ &= (e^{iz} + e^{-iz}) \frac{1}{2} = \cos z \end{aligned}$$

(b)

$$\begin{aligned} \left(\frac{e^z + e^{-z}}{2} \right)^2 - \left(\frac{e^z - e^{-z}}{2} \right)^2 \\ = \frac{1}{4} (e^{2z} + 2 + e^{-2z} - e^{2z} + 2 - e^{-2z}) = 1 \end{aligned}$$

Potenzreihendarstellung:

(a)

$$\begin{aligned} \frac{d}{dz} \sum_{n=0}^{\infty} \frac{(-1)^n z^{2n+1}}{(2n+1)!} &= \sum_{n=0}^{\infty} \frac{(-1)^n (2n+1) z^{2n}}{(2n+1)!} \\ &= \sum_{n=0}^{\infty} \frac{(-1)^n z^{2n}}{(2n)!} = \cos z \end{aligned}$$

(b)


$$\left(\sum_{n=0}^{\infty} \frac{z^{2n}}{(2n)!} \right)^2 - \left(\sum_{n=0}^{\infty} \frac{z^{2n+1}}{(2n+1)!} \right)^2 =$$

$$\sum_{n,k=0}^{\infty} \frac{z^{2n+2k}}{(2n)!(2k)!} - \sum_{n,k=0}^{\infty} \frac{z^{2n+2k+2}}{(2n+1)!(2k+1)!}$$

Die Summe laufen jeweils über alle n und k von 0 bis ∞ .

Stellen Sie sich die Terme als markierte Punkte in der (n,k)-Ebene vor:

```
In[ ]:= ListPlot[Flatten[Table[{n, k}, {n, 0, 6, 1}, {k, 0, 6, 1}], 1]]
```



Statt die Summe über n und k zu führen, können wir auch "diagonal" summieren, also über $i=0,1,\dots$ und für jedes i (als feste Potenz von z) über $n=0,1,\dots,i$ (also $i=n+k$) in der ersten Summe und $n=0,1,\dots,i-1$ (also $i=n+k+1$) in der zweiten.

Den Term mit $n=k=0$ nehmen wir aus der ersten Summe heraus.

$$= 1 + \sum_{i=1}^{\infty} z^{2i} \sum_{n=0}^i \frac{1}{(2n)!(2i-2n)!}$$

$$- \sum_{i=1}^{\infty} z^{2i} \sum_{k=0}^{i-1} \frac{1}{(2n+1)!(2i-2n-1)!}$$

Rechnungsgang mit Lösung.

```
In[ ]:= ClearAll["Global`*"];
```

2.10

Beweisen Sie für beliebige, komplexe z_1, z_2 die Dreiecksungleichung

$$|z_1 + z_2| \leq |z_1| + |z_2|$$


und veranschaulichen Sie diese graphisch.

Lösungsweg

Die beiden komplexen Zahlen (beziehungsweise die Verbindungslinien zum Ursprung) kann man sich als zwei Seiten eines Dreiecks vorstellen. Die Länge der dritten Seite ($|z_1 + z_2|$) ist dann sicher immer kleiner und höchstens gleich der Summe der Längen der beiden anderen Seiten.

Für die beiden komplexen Zahlen $a=1+5i$, $b=-2+3i$ ergibt sich das Bild:

```
In[ ]:= Show[Graphics[
  {Line[{{0, 0}, {1, 5}}], Line[{{1, 5}, {-1, 8}}], Line[{{-1, 8}, {0, 0}}]}]]
```



Für den algebraischen Beweis quadrieren wir die beiden Seiten der Ungleichung und betrachten die Differenz.

$$\begin{aligned}
 (|z_1 + z_2|)^2 - (|z_1|)^2 - (|z_2|)^2 - 2|z_1||z_2| \\
 = (x_1 + x_2)^2 + (y_1 + y_2)^2 - x_1^2 - x_2^2 - y_1^2 - y_2^2 \\
 - 2|z_1||z_2| = \\
 2x_1x_2 + 2y_1y_2 - 2|z_1||z_2|
 \end{aligned}$$

Wir müssen zeigen, dass dieser Ausdruck kleiner oder gleich null ist. also die Ungleichung

$$x_1x_2 + y_1y_2 \leq |z_1||z_2|$$

erfüllt. Da die rechte Seite sicher positiv ist, quadrieren wir beide Seiten und erhalten

$$\begin{aligned}
 (x_1x_2 + y_1y_2)^2 &\leq (x_1^2 + y_1^2)(x_2^2 + y_2^2) \\
 x_1^2x_2^2 + y_1^2y_2^2 + 2x_1x_2y_1y_2 \\
 &\leq x_1^2x_2^2 + y_1^2y_2^2 + y_1^2x_2^2 + x_1^2y_2^2 \\
 2x_1x_2y_1y_2 &\leq y_1^2x_2^2 + x_1^2y_2^2 \\
 0 &\leq (x_1y_2 - x_2y_1)^2
 \end{aligned}$$

Die letzte Zeile entspricht der vorhergehenden, wie man durch Quadrieren erkennt. Auch erfüllt sie sicher die Ungleichung, da ein Quadrat positiv oder null sein kann. Der Beweis ist damit abgeschlossen.

```
In[ ]:= ClearAll["Global`*"];
```

2.12

Schreiben Sie die Potenzreihen in z^n (zumindest bis zum 3. Term) für folgende Funktionen hin und bestimmen Sie das Konvergenzgebiet (vgl. Abschnitt 1.3):

$$(a) \frac{e^z}{(1+z)^2}$$

$$(b) \frac{\sin z}{z(1+z)^3}$$

$$(c) \frac{e^z - e^{-z}}{1-z}.$$

Lösungsweg

Das Konvergenzgebiet wird durch die nächste Singularität bestimmt; in allen drei Beispielen ist die Reihe daher zumindest für $|z| < 1$ konvergent, da die Singularitäten bei (a, b) $z = -1$, (c) $z = 1$ sind. Die scheinbare Singularität im Fall (b) bei $z = 0$ hebt sich wegen der Nullstelle des $\sin(z)$ eben dort weg.

Die Reihenentwicklung verläuft analog zu den übliche Taylorreihen in Kap. 1, also zum Beispiel mit Hilfe der dort besprochenen Ableitungsformel. Es gibt aber in diesen Aufgaben auch Abkürzungen durch das Zusammensetzen aus bekannten Teilreihen. In *Mathematica* schließlich gibt es den vorgefertigten Befehl "Series".

(a)

Die Reihen für e^z und $1/(1+z)$ sind bekannt, und wir nennen sie P1 und P2 :

$$P1 = 1 + z + \frac{z^2}{2} \dots$$

$$P2 = 1 - z + z^2 \dots$$

In beiden Fällen haben wir die Terme $O(z^3)$ nicht mehr angeschrieben. Daraus ergibt sich

$$P1.P2.P2 = 1 - z + \frac{3z^2}{2} + O(z^3)$$

Das ist auch schon die Antwort. Zur Kontrolle verwenden wir noch

die eingebaute Mathematica Funktion und berechnen die Reihe bis $O(z^5)$:

```
Series[E^z/(1+z)^2,{z,0,4}]
```

$$1 - z + \frac{3z^2}{2} - \frac{11z^3}{6} + \frac{53z^4}{24} + O[z]^5$$

(b)

Wir ersparen uns die langweilige Differenzierungsübung, und umgehen die Formel von Taylor durch Multiplikation bekannten Reihen. Die Reihe für $\sin(z)/z$ lautet

$$P1 = (z - \frac{z^3}{6} + \dots) / z = 1 - \frac{z^2}{6} + \dots$$

und die für $1/(1+z)$ ist

$$P2 = 1 - z + z^2 \dots$$

Es ergibt sich

$$P1 \cdot P2 \cdot P2 \cdot P2$$

$$\begin{aligned} &= \left(1 - \frac{z^2}{6}\right) (1 - z + z^2) (1 - z + z^2) (1 - z + z^2) \dots \\ &= 1 - 3z + \frac{35z^2}{6} + O(z^3) \end{aligned}$$

Das ist das gesuchte Ergebnis. Zur Kontrolle die Mathematica Funktion:

`In[*]:= Series[Sin[z] / (z (1 + z) ^ 3), {z, 0, 4}]`

$$\text{Out[*]} = 1 - 3z + \frac{35z^2}{6} - \frac{19z^3}{2} + \frac{1681z^4}{120} + O[z]^5$$

(c)

Diese Rechnung kann durch Kenntnis der Reihe für $\sinh(z)$ und $1/(1-z)$ schnell erfolgen und liefert (hier sogar bis $O(z^5)$):

$$2z + 2z^2 + \frac{7z^3}{3} + \frac{7z^4}{3} + O[z]^5$$

`In[*]:= ClearAll["Global`*"];`

2.13

Suchen Sie alle drei Wurzeln von $\sqrt[3]{1}$. Überprüfen Sie, dass das Quadrat jeder der Wurzeln wieder unter den Lösungen zu finden ist. Zeigen Sie, dass die Summe aller drei Wurzeln null ist. Verallgemeinern Sie letzteres auf die n Lösungen von $\sqrt[n]{z}$ (z beliebig komplex).

Lösungsweg

Wir schreiben dazu zuerst den Ausdruck unter der Wurzel in Polarform, also durch Betrag (1) und Argument ($0+2n\pi$) an, wobei wir die Vieldeutigkeit des Arguments explizit machen.

$$z = 1 = 1 \exp(2n\pi i), \quad n = 0, \pm 1, \pm 2, \pm 3,$$

Die dritte Wurzel daraus ist

$$\sqrt[3]{1} = 1^{1/3} = \exp(2n\pi i / 3)$$

Wenn wir die mögliche Werte von n durchgehen sehen wir,

daß nur drei verschiedene Ergebnisse herauskommen,

also nur drei verschiedene komplexe Zahlen die möglichen Lösungen sind :

$$n = 0 : \exp(0) = 1$$

$$n = 1 : \exp(2\pi i / 3) = \cos(2\pi / 3) + i \sin(2\pi / 3)$$

$$= -\frac{1}{2} + i \frac{\sqrt{3}}{2}$$

$$n = 2 : \exp\left(\frac{4\pi i}{3}\right) = \cos\left(\frac{4\pi}{3}\right) + i \sin\left(\frac{4\pi}{3}\right) \\ = -\frac{1}{2} - i \frac{\sqrt{3}}{2}$$

Alle weiteren Werte von n liefern nur einen dieser drei Fälle. Dies sind daher die drei Wurzeln von $1^{1/3}$.

In Mathematica erhalten wir alle drei Lösungen durch Lösung des Gleichungssystem

$$z^3 = 1$$

und Berechnung von Realteil und Imaginärteil diese Lösungen:

```
In[ ]:= Lösungen = Solve[z^3 == 1, z]
```


```
Out[ ]:= {{z -> 1}, {z -> -(-1)^(1/3)}, {z -> (-1)^(2/3)}}
```

```
In[ ]:= ComplexExpand[z /. Lösungen]
```

```
Out[ ]:= {1, -1/2 - i*sqrt(3)/2, -1/2 + i*sqrt(3)/2}
```

Wir stellen diese drei Lösungen in der komplexen Ebene dar:

```
In[ ]:= Show[Graphics[{Circle[{0, 0}, 1],
  AbsolutePointSize[10], Point[{1, 0}],
  Point[{-1/2, sqrt(3)/2}], Point[{-1/2, -sqrt(3)/2}]}], AspectRatio -> 1]
```



Wir sehen, dass die drei Wurzel äquidistant im Winkel auf dem Kreis liegen.

Das jede der Wurzel die Form $\exp(2n\pi/3)$ ($n=0,1,2$) habe, sind deren Quadrate $\exp(4n\pi/3)$ identisch mit den Lösungen zu $n=0, 2$ und 1 , wenn man die Periodizität (Periode 2π) berücksichtigt.

Man erkennt schon an der Skizze, dass in der Summe aller drei Wurzeln sich sowohl Imaginär- als auch Realteil zu null wegheben.

Diese letzte Aussage kann man auch für beliebige n-te Wurzeln aus komplexen Zahlen zeigen. Auch diese liegen ja auf Kreisen um den Ursprung, wobei der Radius durch den Betrag der n-ten Wurzel aus dem $|z|$ gegeben ist. Die Summe aller n Wurzeln hat dann die Form

$$\sum_{k=0}^{n-1} \exp(2 k \pi i / n) = \sum_{k=0}^{n-1} w^k$$

wobei $w = \exp(2 \pi i / n)$ ist.

Es ist aber

$$\sum_{k=0}^{n-1} w^k = 1 + w + w^2 + \dots + w^{n-1} = \frac{1 - w^n}{1 - w} = 0$$

daja $w^n = 1$ gilt!

```
In[ ]:= ClearAll["Global`*"];
```

2.14

Finden Sie alle Wurzeln, und stellen Sie diese graphisch dar:

(a) \sqrt{i} (b) $\sqrt[3]{i}$ (c) $\sqrt[3]{-8 i}$
 (d) $(-2 \sqrt{3} - 2 i)^{1/4}$ (e) $\sqrt[5]{32}$
 (f) $\sqrt[5]{-1 - i}$ (g) $\sqrt[6]{1}$ (h) $\sqrt[6]{-1}$

Lösungsweg

Die Vorgangsweise bei der Rechnung wurde schon in Aufgabe 2.13 genau besprochen. Hier wollen wir nur mehr die Wurzeln (mit Hilfe der Mathematica Funktionen bestimmen und) graphisch darstellen.

Um eine Lösungsmenge von komplexen Zahle als Punkte darzustellen, schreiben wir uns zuerst ein kurzes Mathematica Programm:

```
In[ ]:= PlotSolutions[Lsg_] := Block[{Rad, Pairs, PTab},
  Rad = Abs[Lsg[[1]]];
  (* = der Radius des Kreises *)
  Pairs = Transpose[{Re[Lsg], Im[Lsg]}];
  PTab = Point/@ Pairs;
  (* ..aus Paare werden Punkt *)
  Show[Graphics[{Circle[{0, 0}, Rad],
    AbsolutePointSize[10], PTab}], AspectRatio -> 1];];
```

(a)

```
In[ ]:= Lsg = ComplexExpand[z /. Solve[z^2 == I, z]]
```

$$\text{Out[]:= } \left\{ -\frac{1+i}{\sqrt{2}}, \frac{1+i}{\sqrt{2}} \right\}$$

```
In[ ]:= PlotSolutions[Lsg];
```

(b)

In[*]:= Lsg = ComplexExpand[z /. Solve[z^3 == I, z]]

$$\text{Out[*]} = \left\{ -i, \frac{i}{2} + \frac{\sqrt{3}}{2}, \frac{i}{2} - \frac{\sqrt{3}}{2} \right\}$$

In[*]:= PlotSolutions[Lsg];

(c)

In[*]:= Lsg = ComplexExpand[z /. Solve[z^3 == -8 I, z]]

$$\text{Out[*]} = \{ 2i, -i - \sqrt{3}, -i + \sqrt{3} \}$$

In[*]:= PlotSolutions[Lsg];

(d)

In[*]:= Lsg = ComplexExpand[z /. Solve[z^4 == -2 Sqrt[3] - 2 I, z]]

$$\text{Out[*]} = \left\{ -\sqrt{2} \cos\left[\frac{5\pi}{24}\right] + i\sqrt{2} \sin\left[\frac{5\pi}{24}\right], -i\sqrt{2} \cos\left[\frac{5\pi}{24}\right] - \sqrt{2} \sin\left[\frac{5\pi}{24}\right], \right. \\ \left. i\sqrt{2} \cos\left[\frac{5\pi}{24}\right] + \sqrt{2} \sin\left[\frac{5\pi}{24}\right], \sqrt{2} \cos\left[\frac{5\pi}{24}\right] - i\sqrt{2} \sin\left[\frac{5\pi}{24}\right] \right\}$$

In[*]:= PlotSolutions[Lsg];

(e)

In[*]:= Lsg = ComplexExpand[z /. Solve[z^5 == 32, z]]

$$\text{Out[*]} = \left\{ 2, -\frac{1}{2} - \frac{\sqrt{5}}{2} - 2i \sqrt{\frac{5}{8} - \frac{\sqrt{5}}{8}}, -\frac{1}{2} + \frac{\sqrt{5}}{2} + 2i \sqrt{\frac{5}{8} + \frac{\sqrt{5}}{8}}, \right. \\ \left. -\frac{1}{2} + \frac{\sqrt{5}}{2} - 2i \sqrt{\frac{5}{8} + \frac{\sqrt{5}}{8}}, -\frac{1}{2} - \frac{\sqrt{5}}{2} + 2i \sqrt{\frac{5}{8} - \frac{\sqrt{5}}{8}} \right\}$$

Das ist zwar richtig, aber verblüffend. Wir hätten als Wurzeln ja

$2 (\cos[2k\pi/5] + i \sin[2k\pi/5])$, $k = 0, 1, 2, 3, 4$

erwartet. Offenbar hat Mathematica die Winkelfunktionen für $2\pi/5$ durch die angegebenen Wurzeln ausgedrückt:

In[*]:= Cos[2 π / 5]

$$\text{Out[*]} = \frac{1}{4} (-1 + \sqrt{5})$$

In[*]:= PlotSolutions[Lsg];

(f)

In[*]:= Lsg = ComplexExpand[z /. Solve[z^5 == -1 - I, z]]

$$\begin{aligned}
 \text{Out[*]} = & \left\{ 2^{1/10} \cos\left[\frac{3\pi}{20}\right] - i 2^{1/10} \sin\left[\frac{3\pi}{20}\right], \right. \\
 & -\frac{\cos\left[\frac{3\pi}{20}\right]}{2 \times 2^{9/10}} - \frac{\sqrt{5} \cos\left[\frac{3\pi}{20}\right]}{2 \times 2^{9/10}} - 2^{1/10} \sqrt{\frac{5}{8} - \frac{\sqrt{5}}{8}} \sin\left[\frac{3\pi}{20}\right] + \\
 & i \left(-2^{1/10} \sqrt{\frac{5}{8} - \frac{\sqrt{5}}{8}} \cos\left[\frac{3\pi}{20}\right] + \frac{\sin\left[\frac{3\pi}{20}\right]}{2 \times 2^{9/10}} + \frac{\sqrt{5} \sin\left[\frac{3\pi}{20}\right]}{2 \times 2^{9/10}} \right), \\
 & -\frac{\cos\left[\frac{3\pi}{20}\right]}{2 \times 2^{9/10}} + \frac{\sqrt{5} \cos\left[\frac{3\pi}{20}\right]}{2 \times 2^{9/10}} + 2^{1/10} \sqrt{\frac{5}{8} + \frac{\sqrt{5}}{8}} \sin\left[\frac{3\pi}{20}\right] + \\
 & i \left(2^{1/10} \sqrt{\frac{5}{8} + \frac{\sqrt{5}}{8}} \cos\left[\frac{3\pi}{20}\right] + \frac{\sin\left[\frac{3\pi}{20}\right]}{2 \times 2^{9/10}} - \frac{\sqrt{5} \sin\left[\frac{3\pi}{20}\right]}{2 \times 2^{9/10}} \right), \\
 & -\frac{\cos\left[\frac{3\pi}{20}\right]}{2 \times 2^{9/10}} + \frac{\sqrt{5} \cos\left[\frac{3\pi}{20}\right]}{2 \times 2^{9/10}} - 2^{1/10} \sqrt{\frac{5}{8} + \frac{\sqrt{5}}{8}} \sin\left[\frac{3\pi}{20}\right] + \\
 & i \left(-2^{1/10} \sqrt{\frac{5}{8} + \frac{\sqrt{5}}{8}} \cos\left[\frac{3\pi}{20}\right] + \frac{\sin\left[\frac{3\pi}{20}\right]}{2 \times 2^{9/10}} - \frac{\sqrt{5} \sin\left[\frac{3\pi}{20}\right]}{2 \times 2^{9/10}} \right), \\
 & -\frac{\cos\left[\frac{3\pi}{20}\right]}{2 \times 2^{9/10}} - \frac{\sqrt{5} \cos\left[\frac{3\pi}{20}\right]}{2 \times 2^{9/10}} + 2^{1/10} \sqrt{\frac{5}{8} - \frac{\sqrt{5}}{8}} \sin\left[\frac{3\pi}{20}\right] + \\
 & i \left(2^{1/10} \sqrt{\frac{5}{8} - \frac{\sqrt{5}}{8}} \cos\left[\frac{3\pi}{20}\right] + \frac{\sin\left[\frac{3\pi}{20}\right]}{2 \times 2^{9/10}} + \frac{\sqrt{5} \sin\left[\frac{3\pi}{20}\right]}{2 \times 2^{9/10}} \right) \left. \right\}
 \end{aligned}$$

Auch hier gilt das unter (e) Gesagte: diese Lösung ist verblüffend, da wir eigentlich das Ergebnis

$$\begin{aligned}
 & 2^{1/10} \left(\cos\left[\frac{\pi}{4} + \frac{2k\pi}{5}\right] + i \sin\left[\frac{\pi}{4} + \frac{2k\pi}{5}\right] \right) \\
 & = 2^{1/10} \left(\cos\left[\frac{(5+8k)\pi}{20}\right] + i \sin\left[\frac{(5+8k)\pi}{20}\right] \right)
 \end{aligned}$$

für $k = 0, 1, 2, 3, 4$

erwartet haben. Mathematica hat Winkeladditionstheoreme verwendet, um alle Winkelfunktionen auf diese zwei

$$\cos\left[\frac{3\pi}{20}\right] \text{ und } \sin\left[\frac{3\pi}{20}\right]$$

zurückzuführen. Wir vergleichen numerisch. Mathematica lieferte:

In[*]:= N[Lsg]

$$\begin{aligned}
 \text{Out[*]} = & \{0.954957 - 0.486575 i, -1.05858 - 0.167662 i, \\
 & 0.757858 + 0.757858 i, -0.167662 - 1.05858 i, -0.486575 + 0.954957 i\}
 \end{aligned}$$

Unser Lösungsvorschlag ergibt:

```
In[*]:= N[Table[21/10 (Cos[ $\frac{(5+8k)\pi}{20}$ ] + I Sin[ $\frac{(5+8k)\pi}{20}$ ]), {k, 0, 4}]]
```

```
Out[*]:= {0.757858 + 0.757858 i, -0.486575 + 0.954957 i,
-1.05858 - 0.167662 i, -0.167662 - 1.05858 i, 0.954957 - 0.486575 i}
```

Beide stimmen also tatsächlich überein, obwohl sich keine Ähnlichkeit erkennen lässt!

```
In[*]:= PlotSolutions[Lsg];
```

(g)

```
In[*]:= Lsg = ComplexExpand[z /. Solve[z^6 == 1, z]]
```

```
Out[*]:= {-1, 1, - $\frac{1}{2} - \frac{i\sqrt{3}}{2}$ ,  $\frac{1}{2} + \frac{i\sqrt{3}}{2}$ ,  $\frac{1}{2} - \frac{i\sqrt{3}}{2}$ , - $\frac{1}{2} + \frac{i\sqrt{3}}{2}$ }
```

```
In[*]:= PlotSolutions[Lsg];
```

(h)

```
In[*]:= Lsg = ComplexExpand[z /. Solve[z^6 == -1, z]]
```

```
Out[*]:= {-i, i, - $\frac{i}{2} - \frac{\sqrt{3}}{2}$ ,  $\frac{i}{2} + \frac{\sqrt{3}}{2}$ , - $\frac{i}{2} + \frac{\sqrt{3}}{2}$ ,  $\frac{i}{2} - \frac{\sqrt{3}}{2}$ }
```

```
In[*]:= PlotSolutions[Lsg];
```

Rechnungsgang mit Lösung.

```
In[*]:= ClearAll["Global`*"];
```

2.16

Sind $(z^{z_1})^{z_2}$ und $z^{z_1 z_2}$ gleich? (Beispiel!)

Lösungsweg

Die Exponentialdarstellung $z = \exp(\log(z))$ zeigt es:

$$\begin{aligned} (z^{z_1})^{z_2} &= (\exp(z_1 \log(z)))^{z_2} = \exp(z_1 z_2 \log(z)) \\ &= \exp(\log(z^{z_1 z_2})) = z^{z_1 z_2} \end{aligned}$$

Beispiel:

```
In[*]:= z = 1 + I;
=  $\sqrt{2} E^{I\pi/4}$ ;
```

```
In[*]:= z1 = 1 + 2 I;
z2 = -3 - I;
z1 z2
```

```
Out[*]:= -1 - 7 i
```

Es muss daher gelten

$$(1 + I)^{-1-7I} = ((1 + I)^{1+2I})^{-3-I}$$

(1) Linke Seite:

$$(1 + I)^{-1-7I} = \text{Exp} [(-1 - 7 I) \text{Log}[1 + I]]$$

In[*]:= **ComplexExpand** [**Log** [**1 + I**]]

$$\text{Out[*]} = \frac{i \pi}{4} + \frac{\text{Log}[2]}{2}$$

In[*]:= **MyExponent** = **Expand** [**(-1 - 7 I)** $\left(\frac{i \pi}{4} + \frac{\text{Log}[2]}{2} \right)$]

$$\text{Out[*]} = \left(\frac{7}{4} - \frac{i}{4} \right) \pi - \left(\frac{1}{2} + \frac{7i}{2} \right) \text{Log}[2]$$

In[*]:= **Re** [**MyExponent**]

$$\text{Out[*]} = \frac{7 \pi}{4} - \frac{\text{Log}[2]}{2}$$

Der Betrag des Ergebnisses $\text{Exp}[\text{MyExponent}]$ ist also $E^{7\pi/4/\sqrt{2}}$.

In[*]:= **Im** [**MyExponent**]

$$\text{Out[*]} = -\frac{\pi}{4} - \frac{7 \text{Log}[2]}{2}$$

Um das Ergebnis als Real- und Imaginärteil darzustellen, verwenden wir hier (nach der Exponentiation) die Mathematica-Funktion **ComplexExpand**.

In[*]:= **Resultat** = **ComplexExpand** [**Exp** [**MyExponent**]]

$$\text{Out[*]} = \frac{e^{7\pi/4} \text{Cos} \left[\frac{\pi}{4} + \frac{7 \text{Log}[2]}{2} \right]}{\sqrt{2}} - \frac{i e^{7\pi/4} \text{Sin} \left[\frac{\pi}{4} + \frac{7 \text{Log}[2]}{2} \right]}{\sqrt{2}}$$

(2) Rechte Seite:

Wir berechnen nun $z_1 z_2 \text{Log}(z)$:

In[*]:= **(1 + 2 I)** $\left(i \pi / 4 + \text{Log}[\sqrt{2}] \right)$

$$\text{Out[*]} = (1 + 2 i) \left(\frac{i \pi}{4} + \frac{\text{Log}[2]}{2} \right)$$

$$z^{\wedge} z_1 = \text{Exp} \left[(1 + 2 I) \left(i \pi / 4 + \text{Log}[\sqrt{2}] \right) \right]$$

$$= \text{Exp} \left[\text{Log}[\sqrt{2}] - \pi / 2 + I (\pi / 4 + \text{Log}[2]) \right]$$

Rechnungsgang mit Lösung.

In[*]:= **ClearAll** ["Global`*"] ;

2.17

Wie lauten Real- und Imaginärteil von

$$z = (1 + \sqrt{3} i)^{10} + (1 - \sqrt{3} i)^{10} ?$$

Lösungsweg

Der Imaginärteil ist sicher 0, da es sich ja um eine Form $a + \bar{a}$ handelt.

Zur Bestimmung des Realteils wählen wir die Polarform.

$$w = 1 + \sqrt{3} i = 2 E^{i \operatorname{Arctan} \sqrt{3}} = 2 E^{i\pi/3}$$

$$w^{10} = 1024 E^{i10\pi/3} = 1024 E^{i4\pi/3}$$

$$z = 2048 \operatorname{Cos} (4 \pi / 3) = -1024$$

`In[]:= ClearAll["Global`*"];`

2.18

Berechnen Sie im Komplexen:

(a) $\sqrt[i]{i}$

(b) $\ln(i e)$

(c) $i \exp\left(1 + \frac{i\pi}{2}\right)$

Lösungsweg

Diese Aufgaben löst man am besten in der Exponentialdarstellung.

(a)

Das ist ein "Klassiker" unter den Aufgaben. Wir schreiben um:

$$\sqrt[i]{i} = (i)^{1/i} = \exp\left(i \left(\frac{\pi}{2} + 2 n\pi\right) / i\right) = \exp\left(\frac{\pi}{2} + 2 n\pi\right)$$

mit $n=0, \pm 1, \pm 2$ usw.

(b)

Es ist

$$\begin{aligned} \ln[i e] &= \ln\left[e \cdot \exp\left(i \left(\frac{\pi}{2} + 2 n\pi\right)\right)\right] \\ &= \ln\left[\exp\left(1 + i \left(\frac{\pi}{2} + 2 n\pi\right)\right)\right] \\ &= 1 + i \left(\frac{\pi}{2} + 2 n\pi\right) \end{aligned}$$

(c)

Wir formen um:

$$\begin{aligned} i \exp\left(1 + \frac{i\pi}{2}\right) &= \exp\left(\frac{i\pi}{2} + 1 + \frac{i\pi}{2}\right) \\ &= \exp(1 + i\pi) = -e \end{aligned}$$

Dabei brauchten wir die Vieldeutigkeit des Arguments nicht zu berücksichtigen da das Ergebnis davon hier nicht abhängt.

`In[]:= ClearAll["Global`*"];`

2.19

Zeigen Sie, dass die stereographische Projektion (M.2.3.1) den Punkt (x,y) in der Ebene auf den

Punkt (X,Y,Z) auf der Kugelfläche projiziert.

Lösungsweg

Die Projektion lautet

$$\begin{aligned} \text{In[*]} &:= \text{XK} = x / (1 + x^2 + y^2); \\ &\text{YK} = y / (1 + x^2 + y^2); \\ &\text{ZK} = (x^2 + y^2) / (1 + x^2 + y^2); \end{aligned}$$

Wir überprüfen, ob diese Werte die Gleichung einer Kugeloberfläche (Kugelmittelpunkt: (0,0,c)) erfüllen:

$$\begin{aligned} \text{In[*]} &:= \text{Simplify}[\text{XK}^2 + \text{YK}^2 + (\text{ZK} - c)^2] \\ \text{Out[*]} &:= \frac{x^2 + y^2 - 2c(x^2 + y^2) + c^2(1 + x^2 + y^2)}{1 + x^2 + y^2} \end{aligned}$$

Für $c = \frac{1}{2}$ ist das offenbar der Fall:

$$\begin{aligned} \text{In[*]} &:= \text{Simplify}[\text{XK}^2 + \text{YK}^2 + (\text{ZK} - \frac{1}{2})^2] \\ \text{Out[*]} &:= \frac{1}{4} \end{aligned}$$

Der Radius ist also ebenfalls $\frac{1}{2}$.

Zum Abschluss einige Beispiele:

$$\begin{aligned} \text{In[*]} &:= \{\text{XK}, \text{YK}, \text{ZK}\} /. \{x \rightarrow 0, y \rightarrow 0\} \\ \text{Out[*]} &:= \{0, 0, 0\} \end{aligned}$$

$$\begin{aligned} \text{In[*]} &:= \{\text{XK}, \text{YK}, \text{ZK}\} /. \{x \rightarrow 1, y \rightarrow 0\} \\ \text{Out[*]} &:= \left\{ \frac{1}{2}, 0, \frac{1}{2} \right\} \end{aligned}$$

$$\begin{aligned} \text{In[*]} &:= \{\text{XK}, \text{YK}, \text{ZK}\} /. \{x \rightarrow 0, y \rightarrow 100\} \\ \text{Out[*]} &:= \left\{ 0, \frac{100}{10001}, \frac{10000}{10001} \right\} \end{aligned}$$

$$\begin{aligned} \text{In[*]} &:= \text{Limit}[\{\text{XK}, \text{YK}, \text{ZK}\} /. \{x \rightarrow 0\}, y \rightarrow \text{Infinity}] \\ \text{Out[*]} &:= \{0, 0, 1\} \end{aligned}$$

$$\begin{aligned} \text{In[*]} &:= \{\text{XK}, \text{YK}, \text{ZK}\} /. \{x \rightarrow 1, y \rightarrow 1\} \\ \text{Out[*]} &:= \left\{ \frac{1}{3}, \frac{1}{3}, \frac{2}{3} \right\} \end{aligned}$$

$$\text{In[*]} := \text{ClearAll}["Global`*"];$$

2.20

Diskutieren Sie die Abbildung $w = \sinh\{z\}$, insbesondere mögliche Mehrdeutigkeiten und Riemannsche Blätter.

Lösungsweg

Wir zerlegen zuerst die Funktion mit $z=x+iy$ und mit Hilfe der Winkeladditionstheoreme in den reellen und imaginären Anteil;

$$\begin{aligned} w &= \sinh(x + iy) \\ &= \sinh(x) \cos(y) + i \cosh(x) \sin(y) \end{aligned}$$

In Mathematica funktioniert da übrigens mit dem Befehl

```
In[ ]:= TrigExpand[Sinh[x + I y]]
```

```
Out[ ]:= i Cosh[x] Sin[y] + Cos[y] Sinh[x]
```

Nun untersuchen wir, welche Wert von $z=x+iy$ auf welche Werte von w abgebildet werden.

Für konstante Werte von x sehen wir aus der obigen Form, dass in der w -Ebene Ellipsen entstehen, deren Achsenabschnitte durch $\sinh(x)$ und $\cosh(x)$ gegeben sind. Obwohl formal die y -Periode 2π ist, werden wir gleich sehen, dass es ausreicht, nur einen Streifen $-\pi/2 \leq y \leq \pi/2$ der z -Ebene zu betrachten, da dieser in die gesamte w -Ebene abgebildet wird.

```
In[ ]:= x = 0.05;
```

```
PYR = ParametricPlot[{Sinh[x] Cos[y], Cosh[x] Sin[y]},  
  {y, -Pi/2, Pi/2}, PlotRange -> {{-1, 1}, {-1, 1}}];
```

Die linke Hälfte der Ellipse entsteht durch

```
In[ ]:= x = -0.05;
```


```
PYL = ParametricPlot[{Sinh[x] Cos[y], Cosh[x] Sin[y]},  
  {y, -Pi/2, Pi/2}, PlotRange -> {{-1, 1}, {-1, 1}}];
```

Es handelt sich hier also um eine sehr schmale (in vertikale Richtung) orientierte Ellipse.

Wir betrachten noch zwei x -Werte und zeichnen dann die Ellipsen in der w -Ebene gemeinsam.

```
In[ ]:= P1 = {PYL, PYR, ParametricPlot[{Sinh[-1] Cos[y], Cosh[-1] Sin[y]},  
  {y, -Pi/2, Pi/2}, DisplayFunction -> Identity],  
  ParametricPlot[{Sinh[1] Cos[y], Cosh[1] Sin[y]},  
  {y, -Pi/2, Pi/2}, DisplayFunction -> Identity}};
```

```
In[ ]:= Show[P1, DisplayFunction -> $DisplayFunction,
  PlotRange -> All, AspectRatio -> 1]
```




In der z -Ebene sind dies die vertikalen Strecken $x=-1, -0.05, 0.05, 1$ und $-\pi/2 \leq y \leq \pi/2$.

Die Kurven umlaufen alle die Strecke $(-i, i)$ der imaginären w -Achse. Im Grenzfall $x=0$ liegen sie auf dieser Strecke. Offenbar spielen die Punkte $w = i, -i$ eine zentrale Rolle.

Wir betrachten nun noch die in der z -Ebene horizontalen Geraden $y=-1.4, -1, -0.1, 0.1, 1, 1.4$:


```
In[ ]:= P2 = {ParametricPlot[{Sinh[x] Cos[-1.4], Cosh[x] Sin[-1.4]},
  {x, -2, 2}, DisplayFunction -> Identity],
  ParametricPlot[{Sinh[x] Cos[-1], Cosh[x] Sin[-1]},
  {x, -2, 2}, DisplayFunction -> Identity],
  ParametricPlot[{Sinh[x] Cos[-0.1], Cosh[x] Sin[-0.1]},
  {x, -2, 2}, DisplayFunction -> Identity],
  ParametricPlot[{Sinh[x] Cos[0.1], Cosh[x] Sin[0.1]},
  {x, -2, 2}, DisplayFunction -> Identity],
  ParametricPlot[{Sinh[x] Cos[1], Cosh[x] Sin[1]},
  {x, -2, 2}, DisplayFunction -> Identity],
  ParametricPlot[{Sinh[x] Cos[1.4], Cosh[x] Sin[1.4]},
  {x, -2, 2}, DisplayFunction -> Identity]};
```

```
In[ ]:= Show[P2, DisplayFunction -> $DisplayFunction,
  PlotRange -> All, AspectRatio -> 1]
```



Das sind Hyperbeln: für negative y in der unteren, für positive y in der oberen Halbebene. Zum Abschluss noch das Gesamtbild:

```
In[ ]:= Show[P1, P2, DisplayFunction -> $DisplayFunction,
  PlotRange -> All, AspectRatio -> 1]
```



Wir sehen: $w(z)$ ist periodisch in Bezug auf y ; jeder π breite Streifen parallel zur x -Achse wird auf die gesamte w -Ebene abgebildet.

Eine besondere Rolle spielen die Punkte $w=i, -i$, wie man auch an der folgenden Umkehrung der

Funktion erkennen kann.

$$w = \text{Sinh}[z]$$

$$\rightarrow 2w = e^z - e^{-z}$$

$$\rightarrow e^{2z} - 2we^z - 1 = 0$$

$$\rightarrow e^z = w \pm \sqrt{1 + w^2}$$

$$\rightarrow z = \ln\left[w \pm \sqrt{1 + w^2}\right]$$


Diese Umkehrfunktion zeigt also die beobachtete Vielblättrigkeit: entsprechend der Vielblättrigkeit des Logarithmus, multipliziert mit der Zweiblättrigkeit der Quadratwurzel ergeben sich die diskutierten Streifen der Breite π in der z -Ebene. Die Nullstellen der Wurzel-Diskriminante $w = \pm i$ sind also die Verzweigungspunkte in der w -Ebene. Weitere Verzweigungspunkte liegen im Unendlichen. So können wir die Schnitte der w -Ebene auf die imaginäre Achse legen: $(-\infty, -i)$, (i, ∞) .

Betrachten wir die Projektion eines Halbkreises in der z -Ebene (zum Beispiel der untere Halbkreis um $z = i\pi/2$) in die w -Ebene.

```
In[ ]:= x = 0.1 Sin[t];
```

```
y = Pi / 2 + 0.1 Cos[t];
```


```
In[ ]:= ParametricPlot[ {Sinh[x] Cos[y], Cosh[x] Sin[y]},
  {t, -Pi / 2 + .05, Pi / 2 - 0.05}, AspectRatio -> 1]
```



Wir haben dabei den Halbkreis nicht voll ausgeführt, um darzustellen, dass der entsprechende Kreis in der w -Ebene dann oben (wo wir den Schnitt gelegt haben) offen ist. Offenbar wird dort das Blatt gewechselt.

Die Kurve sieht auf den ersten Blick wie ein Kreis aus, ist aber keiner. Man erkennt das, indem man zum Beispiel das Radiusquadrat zeichnet:

```
In[ ]:= Plot[(Sinh[x] Cos[y])^2 + (Cosh[x] Sin[y])^2, {t, -Pi/2 + .05, Pi/2 - 0.05}]
```



Wir fassen zusammen:

Die w -Ebene ist vielblättrig, man kann dies durch zwei Schnitte $(-\infty, -i)$, (i, ∞) darstellen. Wenn man in w einen Weg durch einen der Schnitte führt, wechselt man das Blatt. Dieser Vorgang entspricht in z einem Übergang in einen benachbarten Streifen.

```
In[ ]:= ClearAll["Global`*"];
```

2.21

Man diskutiere die Schnittstruktur von

(a) $f(z) = \ln\left(\frac{z-a}{z-b}\right)$,

(b) $f(z) = \ln[(z-a)(z-b)]$.

Gibt es Wege, die beide Verzweigungspunkte umrunden und die dennoch geschlossen sind?
Kann man einen Schnitt zwischen die beiden Verzweigungspunkte legen?

Lösungsweg

Zur Analyse bringt man am besten die Funktionen in die Form

$$w(z) = \ln(z-a) \pm \ln(z-b).$$

Die Punkte $z=a$ und $z=b$ sind sicher Verzweigungspunkte. Wir werden in beiden Fällen Kreise in der z -Ebene um $z=0$, a und b , sowie um beide Punkte parametrisieren und deren Projektion in die w -Ebene betrachten.

Achtung:

Der Schnitt der Funktion $\text{Log}[z]$ liegt in Mathematica auf der negativen reellen Achse $(-\infty, 0)$. Dies muss bei der Verwendung dieser Funktion berücksichtigt werden.

Beispiel:

Bei $z=-1$ gibt es eine Diskontinuität

```
In[ ]:= Log[-1 - I 0.01]
```

```
Out[ ]:= 0.0000499975 - 3.13159 i
```

```
In[*]:= Log[-1 + I 0.01]
```

```
Out[*]= 0.0000499975 + 3.13159 i
```

```
In[*]:= Limit[Log[-1 + I y] - Log[-1 - I y], y -> 0.000001]
```

```
Out[*]= 0. + 6.28318 i
```

Bei $z=1$ gibt es keine Diskontinuität

```
In[*]:= Log[1 - I 0.01]
```

```
Out[*]= 0.0000499975 - 0.00999967 i
```

```
In[*]:= Log[1 + I 0.01]
```

```
Out[*]= 0.0000499975 + 0.00999967 i
```

```
In[*]:= Limit[Log[1 + I y] - Log[1 - I y], y -> 0.000001]
```

```
Out[*]= 0. + 2. × 10-6 i
```

Zur numerischen Diskussion wählen wir $a=-a$, $b=1$. Die Funktion ist damit

$$w(z) = \ln(z+1) \pm \ln(z-1).$$

Die folgenden Kreise umlaufen die Punkte 0 , $z=a=-1$, $z=b=1$, sowie $z=a$ und $z=b$.


```
In[*]:= Clear[z, z0, za, zb, x, y]
```

```
In[*]:= z0[t_] := 0.1 (Cos[t] + I Sin[t]);
za[t_] := -1 + 0.1 (Cos[t] + I Sin[t]);
zb[t_] := 1 + 0.1 (Cos[t] + I Sin[t]);
zab[t_] := 1.1 (Cos[t] + I Sin[t]);
```


(a)

```
In[*]:= w[z_] = N[Log[z + 1] - Log[z - 1]];
```


```
In[ ]:= ParametricPlot[{Re[w[za[t]]], Im[w[za[t]]]}, {t, 0, 2 Pi}]
```




```
In[ ]:= ParametricPlot[{Re[w[zb[t]]], Im[w[zb[t]]]}, {t, 0, 2 Pi}]
```



```
In[ ]:= ParametricPlot[{Re[w[z0[t]]], Im[w[z0[t]]]}, {t, 0, 2 Pi}]
```



```
In[ ]:= ParametricPlot[{Re[w[zab[t]]], Im[w[zab[t]]]}, {t, 0, 2 Pi}]
```



Jeder der ersten drei Kreiswege hat also eine Diskontinuität beinhaltet, überschritt (in z) also einen Schnitt. Nur der letzte Kreis führt auch in w zu einer kontinuierlichen, geschlossenen Kurve. Wir schließen daraus, dass wir (in der z -Ebene) einen Schnitt zwischen a und b legen können, in unserem Beispiel also zwischen $(-1,1)$.


Tatsächlich ergibt auch die analytische Betrachtung, dass sich zum Beispiel auf der positiven Achse im Bereich $(1, \infty)$ die Diskontinuitäten wegheben. Die spezielle Wahl des Schnittes (der a und b verbindet, aber im Prinzip ja beliebig gelegt werden kann) ergibt sich in unserer Rechnung aus der (in Mathematica vorgegebenen) Wahl des Schnittes von $\ln(z)$.

(b)


```
In[ ]:= w[z_] = N[Log[z + 1] + Log[z - 1]];
```

Hier werden sich die Diskontinuitäten nicht wegheben.


```
In[ ]:= ParametricPlot[{Re[w[za[t]]], Im[w[za[t]]]}, {t, 0, 2 Pi}]
```




```
In[ ]:= ParametricPlot[{Re[w[z_b[t]]], Im[w[z_b[t]]]}, {t, 0, 2 Pi}]
```



```
In[ ]:= ParametricPlot[{Re[w[z_0[t]]], Im[w[z_0[t]]]}, {t, 0, 2 Pi}]
```



```
In[ ]:= ParametricPlot[{Re[w[zab[t]]], Im[w[zab[t]]]}, {t, 0, 2 Pi}]
```



Jeder der Wege überschreitet also einen Schnitt. Die Schnittstruktur weist hier zwei Schnitte in den Bereichen $(-\infty, -1)$ und $(-\infty, 1)$ auf, also zum Teil übereinander liegend. Wieder ist diese Wahl durch die Konvention der Lage des Schnittes von $\ln(z)$ gegeben.

Man könnte den Schnitt von $\ln(z+1)$ nach links und den von $\ln(z-1)$ nach rechts verlaufen lassen, dann wäre der Bereich der reellen Achse $(-1, 1)$ ohne Schnitt und der Kreis um 0 diskontinuitätsfrei. (Mathematica allerdings hat uns hier keine Wahl gelassen.)

```
In[ ]:= ClearAll["Global`*"];
```

2.22

Zeigen Sie, dass die Funktion

$$f(z) = z^{\frac{1}{2}} + z^{\frac{1}{3}}$$

sechs Riemannsche Blätter hat.

Lösungsweg

Ein Verzweigungspunkt dieser Funktion ist offenbar 0 (der andere liegt im Unendlichen).

Wenn wir eine Kreisbahn rund um $z=0$ durchlaufen, müssen wir diesen Punkt 6 Mal umrunden, bevor wir wieder am gleichen w -Punkt ankommen.

$$f(z) = z^{\frac{1}{2}} + z^{\frac{1}{3}}$$

$$\varepsilon = 0.1$$

$$z = \varepsilon e^{it} \rightarrow w = \sqrt{\varepsilon} e^{it/2} + \sqrt[3]{\varepsilon} e^{it/3}$$


Wir zeichnen nun die verschiedenen Wegen in $w(z)$, die sich ergeben, wenn man $z=0$ jeweils 1,2,3,4,5,6 mal umrundet.

In[*]:= **eps = 0.025;**


Realteil := N[eps^(1/2) Cos[t/2] + eps^(1/3) Cos[t/3]];

Imaginärteil := N[eps^(1/2) Sin[t/2] + eps^(1/3) Sin[t/3]];


In[*]:= **ParametricPlot[{Realteil, Imaginärteil}, {t, 0, 2 Pi},
PlotRange -> {{-1, 1}, {-1, 1}}, AspectRatio -> 1]**




```
In[ ]:= ParametricPlot[{Realteil, Imaginärteil}, {t, 0, 4 Pi},  
PlotRange -> {{-1, 1}, {-1, 1}}, AspectRatio -> 1]
```




```
In[ ]:= ParametricPlot[{Realteil, Imaginärteil}, {t, 0, 6 Pi},  
PlotRange -> {{-1, 1}, {-1, 1}}, AspectRatio -> 1]
```




```
In[ ]:= ParametricPlot[{Realteil, Imaginärteil}, {t, 0, 8 Pi},  
PlotRange -> {{-1, 1}, {-1, 1}}, AspectRatio -> 1]
```



```
In[ ]:= ParametricPlot[{Realteil, Imaginärteil}, {t, 0, 10 Pi},  
PlotRange -> {{-1, 1}, {-1, 1}}, AspectRatio -> 1]
```



```
In[ ]:= ParametricPlot[{Realteil, Imaginärteil}, {t, 0, 12 Pi},
  PlotRange -> {{-1, 1}, {-1, 1}}, AspectRatio -> 1]
```



Es ist klar: die z-Ebene ist 6-blättrig!

```
In[ ]:= ClearAll["Global`*"];
```

2.23

Untersuchen Sie die Riemann-Blatt-Struktur einiger Funktionen (aus dem Text und aus den Aufgaben weiter oben) mit Hilfe eines Computerprogramms, indem Sie die Position von Punkten und Wegen in z in die entsprechenden Punkte und Wege in w umrechnen. Versuchen Sie so auf graphischem Weg die Lage von Verzweigungspunkten zu bestimmen.

Lösungsweg

Es geht um die Beziehung zwischen (mit t parametrisierten) Wegen $z(t)$ und $w(t)=f(z(t))$.

Im Idealfall würde das so aussehen: Das Computerprogramm stellt am Bildschirm links die w - und rechts die z -Ebene dar. Mit der Maus bewegt man sich beliebig über die z -Ebene. Die jeweilige Position wird vom Programm abgelesen und (entsprechen dem Funktionszusammenhang) in einen Wert in der w -Ebene übersetzt und dort durch einen Punkt dargestellt. Bei Bedarf kann man den in z und w zurückgelegten Weg geeignet (farbig?) markieren. Man kann auch Wege in w (mittels Umkehrung der Funktion) als Wege in z darstellen.

Wenn die z -Ebene aus mehreren Riemannblättern besteht, besteht die eigentliche Komplikation darin, das Computerprogramm so zu gestalten, daß die Vieldeutigkeit geeignet aufgelöst wird. Dazu muss man zuerst beachten, dass die vorgegebenen Funktionsaufrufe immer nur die Werte auf dem ersten Riemann-Blatt liefern. Der Programmierer des Pakets hat sich dabei an eine gegebene (hoffentlich im Handbuch geeignet dokumentierte) Konvention gehalten, wo sich die jeweiligen Schnitte befinden.

Beispiel: Log[z]

hat einen Verzweigungspunkt bei $z=0$; der Schnitt wird meist (und auch in Mathematica) auf die reelle Achse nach links gelegt. Wir werden daher eine Diskontinuität von der Größe $2\pi i$ beobachten, wenn wir Werte knapp über mit solchen knapp unter diesem Schnitt vergleichen.

Bei $z=-10\pm i\epsilon$ gibt es einen Schnitt:

```
In[ ]:= Wertoberhalb = Log[-10 + 0.00001 I]
```

```
Out[ ]:= 2.30259 + 3.14159 i
```

```
In[ ]:= Wertunterhalb = Log[-10 - 0.00001 I]
```

```
Out[ ]:= 2.30259 - 3.14159 i
```

Bei $z=10\pm i\epsilon$ gibt es keinen Schnitt:

```
In[ ]:= Wertoberhalb = Log[10 + 0.00001 I]
```

```
Out[ ]:= 2.30259 + 1. × 10-6 i
```

```
In[ ]:= Wertunterhalb = Log[10 - 0.00001 I]
```

```
Out[ ]:= 2.30259 - 1. × 10-6 i
```

Wenn wir also zum Beispiel eine kreisförmige Bahn um $z=0$ durchlaufen, dann wird $w=\text{Log}[z]$ einen Sprung von $2\pi i$ haben, wenn z den Schnitt überschreitet und man einfach die Computerwerte übernimmt. Um zu sehen, dass der Wert in w kontinuierlich ist, muss man für den "Wechseln von z ins zweite Blatt" sorgen. Das heißt in der Praxis: man muss buchhalten, ob beim Weg in z der Schnitt (und in welche Richtung) überquert wurde, und dann geeignet Vielfache von $2\pi i$ zu $\text{Log}(z)$ addieren oder davon abziehen.

Zusammenhängende (kontinuierliche) Wege in z (oder w) müssen im Endprodukt immer auch kontinuierliche Wege in w (oder z) entsprechen. Die Vielblättrigkeit ist dadurch erkennbar, dass verschiedene Punkte in der einen Variablen dann denselben Punkt (obwohl in verschiedenen Blättern übereinander liegend) in der anderen Variablen ergeben.

```
In[ ]:= ClearAll["Global`*"];
```

Julia-Mengen: Apfelmännchen

Dieses Beispiel zur Computer-Box C.2.1 zeigt einerseits, wie man die berühmte Mandelbrotmenge zeichnen kann, andererseits zeigt es aber auch die Grenzen einer interpretativen Sprache wie Mathematica auf. Die Zeichenauflösung ist hier ziemlich grob, da der Rechenaufwand ziemlich schnell steigt, sowohl mit der Zahl der Punkte als auch mit der Iterationstiefe.

Wir definieren zuerst die Iterationsvorschrift:

$$z_0 = 0, \quad z_{n+1} = z_n^2 + c;$$

(Der erste Schritt ist einfach $z_1=c$.)


```
In[ ]:= Iterate[c_] := Block[{i, z},
  i := 1; z = c;
  While[(((i < 201) && (Abs[z] < 2)),
    z = N[z * z + c];
    i++]; Return[i];]
```

Diese Funktion liefert die Iterationszahl (eine Zahl zwischen 1 und 200) als Ergebnis. Wenn diese =200 ist, dann liegt der Punkt c (näherungsweise) in der Juliamenge.

```
In[ ]:= Julia = Table[Iterate[x + I y], {y, -1.25, 1.25, 0.025}, {x, -2, 0.5, 0.025}];
```

Nun zeichnen wir die Juliamenge.


```
In[ ]:= ListDensityPlot[Julia, Mesh -> False]
```



Die Juliamenge ist hier durch Grautöne dargestellt (waagrecht: x , vertikal: y). Der schwarze Bereich gehört nicht zur Juliamenge.


Farbig dargestellt ist die Graphik noch attraktiver

```
In[*]:= ListDensityPlot[Julia, Mesh -> False, ColorFunction -> Hue]
```



Mit Hilfe einer Streckungstransformation (kleine Iterationszahlen werden in Richtung größere verzerrt) sehen wir noch mehr:

```
In[*]:= JJ = N[Sqrt[Julia/201]];
ListDensityPlot[JJ, Mesh -> False, ColorFunction -> Hue]
```



```
In[*]:= ClearAll["Global`*"];
```